'Initialize Processor A

'Set fuse bit for 16 MHz crystal clock

'The program generates the positive half of the 50Hz sinusoidal waveform

'PWM frequency is 8000Hz

   Dim Control As Byte           'Parallel 8 bit control signal from Control Processor

   Dim Control1 As Single        'Used to calculate sine wave amplitude

   Dim A(81) As Integer          'Arrays for sine wave value storage

   Dim B(81) As Integer          ' - A is for the "on" time and B is for the "off" time

   Dim A1(81) As Single         ‘Reference Sine Wave values storage

   Dim X As Integer              'to control For - Next loops for sine wave calculations

   Dim M As Integer              'to control For - Next loops for sine wave generation

   Dim Y As Single               'Sine Wave Amplitude

   Dim Z As Integer              'Sine Wave Amplitude tempory storage

   Config Portb.0 = Input        '"Proceed" input from Processor B

   Config Portb.1 = Output       'PWM output of positive half wave

   Config Portb.2 = Output       '"Wait" indicator to Processor B

   Config Portd = Input          'Control Word "N" from Control Processor C

      Portb.1 = 1       'disable PWM

      Portb.2 = 1       'Processor B must wait

      Y = 0             'Zero sine wave amplitude

                        '1/50=0.02s, 1/8000=125us, 0.02/0.000125 = 160, half of that = 80

      For X = 0 To 80       'calculate the sinusoidal waveform values

      Y = X * 0.039269908    'convert to radians: 180 deg divided by 80 = 2.25 deg = 0.039269908 radians

      Y = Sin(y)             'Calculate sine wave

      Y=Y*480                ' adjust to get 125us period (8000Hz) with the 16MHz Micro clock speed.

      A1(x + 1) = Y          'store values in an array

      Next X

      Wait 1                'Wait for processors to start up and stabilize

Start:

      Portb.1 = 1           'disable PWM

      Control = Inp(&H30)   'read the voltage control word from the control processor C

      Control1 = Control     'calculate sinusoidal voltage waveform amplitude adjusting factor

      Control1 = .035294118 * Control1     ' 9/255 = .035294118

      Control1 = 10 - Control1

      Portb.2 = 0       'Processor B must proceed

      For X = 1 To 81         'calculate the PWM wait times for the sinusoidal voltage waveform

      Y = A1(x)

      Y = Y / Control1

      Z = Y

      A(x) = Z

      B(x) = 480 - Z        'adjust to get 125us period (8000Hz) with the 16MHz Micro clock speed.

      Next X

      Portb.2 = 1       'Prosessor B must wait

      Bitwait Pinb.0 , Reset      'Proceed when processor B indicator is present

'***********************     'Generate positive half of sine wave

     Waitus 500             'Generate the first quarter of the sine wave

     For M = 1 To 41

     Portb.1 = 0

     waitus A(m)

     Portb.1 = 1

     waitus B(m)

     Next M

     Portb.1 = 0          'add in a centre piece

     waitus A(m)

     For M = 42 To 81       'Generate the second quarter of the sine wave

     Portb.1 = 1

     waitus B(m)

     Portb.1 = 0

     waitus A(m)

     Next M

     Portb.1 = 1         'disable PWM output

     Portb.2 = 0        'Processor B must proceed

